<?xml version="1.0" encoding="UTF-8"?>
<xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform" xmlns:msxsl="urn:schemas-microsoft-com:xslt" version="1.0">
<xsl:output method="html" doctype-system="http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd" doctype-public="-//W3C//DTD XHTML 1.0 Transitional//EN"/>
<xsl:variable name="title" select="/rss/channel/title"/>
<xsl:template match="/">
<xsl:element name="html">
	<body>
		<font face="Arial" size="small">
			<table cellpadding="2" cellspacing="2" bgcolor="#333366">
				<tbody>
					<tr>
						<th align="left">
							<font color="white">
								<xsl:value-of select="$title"/>
							</font>
						</th>
					</tr>
					<tr>
						<td bgcolor="white">
							<xsl:apply-templates select="rss/channel"/>
						</td>
					</tr>
				</tbody>
			</table>
		</font>
	</body>
</xsl:element>
</xsl:template>
<xsl:template match="channel">
	<xsl:apply-templates select="item">
			<!--xsl:sort select="concat(substring(pubDate, 9,2),substring(pubDate, 6,2),substring(pubDate, 6,2))" order="descending"/-->
	</xsl:apply-templates>
</xsl:template>
<xsl:template match="item">
	<table cellpadding="4" cellspacing="0" width="100%">
		<tbody>
			<tr bgcolor="#ccccff">
				<xsl:if test="count(child::enclosure)=1">
					<td valign="middle">
						<img src="{enclosure/@url}"/>
					</td>
				</xsl:if>
				<td valign="middle" width="100%">
					<b>
						<a href="{link}">
							<xsl:value-of select="title"/>
						</a>
						<br/>
						<xsl:if test="count(child::pubDate)=1">
							<font color="#333366">Posted: <xsl:value-of select="pubDate"/></font>
						</xsl:if>
					</b>
				</td>
			</tr>
			<tr>
				<td colspan="2">
					<div class="itemcontent" name="decodeable">
						<xsl:call-template name="outputContent"/>
					</div>
				</td>
			</tr>
		</tbody>
	</table>
</xsl:template>
<xsl:template name="outputContent">
	<xsl:choose>
		<xsl:when xmlns:xhtml="http://www.w3.org/1999/xhtml" test="xhtml:body">
			<xsl:copy-of select="xhtml:body/*"/>
		</xsl:when>
		<xsl:when xmlns:xhtml="http://www.w3.org/1999/xhtml" test="xhtml:div">
			<xsl:copy-of select="xhtml:div"/>
		</xsl:when>
		<xsl:when xmlns:content="http://purl.org/rss/1.0/modules/content/" test="content:encoded">
			<xsl:value-of select="content:encoded" disable-output-escaping="yes"/>
		</xsl:when>
		<xsl:when test="description">
			<xsl:value-of select="description" disable-output-escaping="yes"/>
		<hr size="1" width="95%" style="color:#FF1493"></hr>
		</xsl:when>
	</xsl:choose>
</xsl:template>
</xsl:stylesheet>
